On the Computer Proof of a Result in the Abstract Theory of Segments

Dimiter Skordev

Sofia University, Sofia, Bulgaria

In Y. Tagamlitzki's paper [4] an axiomatization for an abstract notion of segment is proposed (a somewhat more restrictive but similar axiomatization has been given earlier by W. Prenowitz in [2]). The main result of [4] is a separation theorem formulated in set-theoretical terms and proved by using Zorn's Lemma. The results published in [4] have been presented earlier in a lecture course held in 1962 at the Faculty of Mathematics of the Sofia University. An intriguing logical problem arose in one of the lectures when the mentioned separation theorem has been applied to obtain a certain non-trivial corollary expressible without essential use of set-theoretical terminology. We direct now our attention to that logical problem.

Let us consider sentences in a first-order predicate language with a ternary relation symbol r. The reader may intuitively interpret $\mathrm{r}(x,y,z)$ as "z is on the segment with end points x and y", regarding the variables in the sentences below as ranging over the points of a plane; however, there are also other appropriate interpretations. The problem in question can be formulated as follows.

Problem 1 *Consider next four sentences:*

(symm) $\forall x \forall y \forall z (r(x,y,z) \rightarrow r(y,x,z))$

(assoc1) $\forall x \forall y \forall u \forall v \forall z (r(x,y,z) \wedge r(z,v,u) \rightarrow \exists w (r(y,v,w) \wedge r(x,w,u)))$

(assoc2) $\forall x \forall y \forall u \forall v \forall z (r(x,z,y) \wedge r(u,z,v) \rightarrow \exists w (r(x,v,w) \wedge r(u,y,w)))$

(monot) $\forall x \forall y \forall u \forall z (r(x,z,u) \wedge r(y,y,z) \rightarrow r(x,y,u))$

Show that their conjunction implies the sentence

$$\forall x \forall y \forall z (r(z,z,x) \wedge r(z,z,y) \rightarrow \exists w (((w{=}x) \vee r(x,x,w)) \wedge ((w{=}y) \vee r(y,y,w))))$$

The statement at the end of the above problem (after "Show that") is in fact the mentioned corollary, but presented in a formal way. Taking into account Gödel's Completeness Theorem, the people that consider Set Theory reliable ought to believe that the problem must have also a solution formalizable in the first-order predicate calculus. On the other hand, no idea how to actually find such a solution emerged initially. In particular, the attempts of the present author in this direction remained without a success. Yet a solution of the indicated kind (presented in the ordinary mathematical language) was given by Ivan Prodanov shortly afterwards (still in 1962). In fact, Prodanov established a stronger result – he actually gave such a solution to next problem:

Problem 2 *Show that the conjunction of the same four sentences* **(symm)**, **(assoc1)**, **(assoc2)** *and* **(monot)** *implies the sentence*

(meet) $\forall x \forall y \forall z (r(z,z,x) \wedge r(z,z,y) \rightarrow \exists w (r(x,x,w) \wedge r(y,y,w)))$

Unfortunately, the published version of Prodanov's proof (present in [3]) is in a form that establishes only what is stated in Problem 1. At the end of 1983 and the beginning of 1984 Jörg Siekmann and his group in Karlsruhe made attempts to solve Problem 2 by using their Markgraf Karl Refutation Procedure (MKRP) [1]. Since the attempts did not succeed, the problem has been decomposed into the following two other ones (the decomposition has been suggested by the present author on the basis of an analysis of Prodanov's proof).

Problem 3 (quite easy) *Show that the conjunction of* **(assoc1)** *and* **(monot)** *implies the sentence*

(transit) $\forall u \forall x \forall y \forall z\ (r(x,u,y) \wedge r(y,u,z) \rightarrow r(x,u,z))$

Problem 4 (hard enough) *Show that the conjunction of* **(symm)**, **(assoc2)** *and* **(transit)** *implies* **(meet)**.

In May 1984 MKRP succeeded on both Problem 3 and Problem 4. Then we made an analysis of the solutions produced by MKRP and observed that actually a solution of the following variant of Problem 4 is contained in the second of them.

Problem 5 *Consider next sentence:*

(assoc2′) $\forall x \forall y \forall u \forall v \forall z\ (r(x,z,y) \wedge r(u,z,v) \rightarrow \exists w (r(v,x,w) \wedge r(y,u,w)))$

Show that the conjunction of **(assoc2′)** *and* **(transit)** *implies* **(meet)**.

Obviously the conjunction of **(symm)** and **(assoc2)** implies **(assoc2′)**. Therefore the statement of Problem 5 can be considered stronger than the statement of Problem 4. Nevertheless, it seems that Problem 5 is less difficult than Problem 4 for a computer proof search. For example, we succeeded in 1998 to find a computer solution of Problem 5 by using Prolog in an appropriate way. We are now going to describe how this can be done.

First of all, the formula **(assoc2′)** must be transformed into a Skolem normal form. The following formula is a convenient prenex normal form of **(assoc2′)**:

$$\forall x \forall y \forall u \forall v \exists w \forall z (r(x,z,y) \wedge r(u,z,v) \rightarrow r(v,x,w) \wedge r(y,u,w))$$

A corresponding Skolem normal form is

$$\forall x \forall y \forall u \forall v \forall z (r(x,z,y) \wedge r(u,z,v) \rightarrow r(v,x,f(x,y,u,v) \wedge r(y,u,f(x,y,u,v)))$$

Taking this into account, we may do an almost straightforward translation of the statement of Problem 5 in Prolog and thus get the next goal and the following program intended to be executed on it:

```
?- r(x,x,W),r(y,y,W).                              % goal g

r(z,z,x).                                          % clause i
r(z,z,y).                                          % clause j
r(V,X,f(X,Y,U,V)) :-  r(X,Z,Y),r(U,Z,V).           % clause k
r(Y,U,f(X,Y,U,V)) :-  r(X,Z,Y),r(U,Z,V).           % clause l
r(X,U,Z) :-  r(X,U,Y),r(Y,U,Z).                    % clause m
```

An execution of this program on the specified goal, however, will be useless – such an execution usually terminates with a memory overflow message. Therefore we shall execute a more sophisticated program on a more sophisticated goal and they will concern derivability in a formal system **S** that generates the minimal Herbrand model for the above program. The formal system in question has the ground atoms r(z,z,x) and r(z,z,y) in clauses i and j as its axioms. The inference rules of the system correspond to clauses k, l and m and can be formulated in the terms of their ground instances. Namely, the rule corresponding to clause k allows to infer the ground atom $\mathcal{C}$ from the ground atoms $\mathcal{A}$ and $\mathcal{B}$, whenever $\mathcal{C}$:– $\mathcal{A}, \mathcal{B}$ is a ground instance of clause k, and similarly for clauses l and m. To find a solution of Problem 5 formalizable in the first-order predicate calculus, it is sufficient to derive in **S** two ground atoms $\mathcal{A}$ and $\mathcal{B}$ such that ?– $\mathcal{A}, \mathcal{B}$ is an instance of the goal g. In order to organize the search for such two ground atoms, we shall use the notion of a *derivation tree* in **S**, defining such trees as appropriate terms built up from two new constants i and j by means of three new binary function symbols k, l, m. Namely, we accept the convention that the constants i and j are derivation trees for the axioms of **S** in clauses i and j, respectively, we accept further that, whenever $\mathcal{C}$:– $\mathcal{A}, \mathcal{B}$ is a ground instance of clause k and $\mathcal{D}$, $\mathcal{E}$ are derivation trees for the ground atoms $\mathcal{A}$ and $\mathcal{B}$, respectively, then the term $\mathrm{k}(\mathcal{D}, \mathcal{E})$ is a derivation tree for the ground atom $\mathcal{C}$, and similarly for the other two inference rules of **S**. *The height* of a derivation tree can be defined in a natural inductive way – the constants i and j have height 0, and the height of any of the derivation trees $\mathrm{k}(\mathcal{D}, \mathcal{E})$, $\mathrm{l}(\mathcal{D}, \mathcal{E})$, $\mathrm{m}(\mathcal{D}, \mathcal{E})$ is the least integer greater than the heights of both $\mathcal{D}$ and $\mathcal{E}$. For the sake of greater efficiency, we shall represent the natural numbers 0, 1, 2, ... by the lists [], [[]],[[[]]], ..., respectively. Instead of the ternary predicate symbol r we shall now use a 5-ary predicate symbol d. Its interpretation can be explained as follows: $\mathrm{d}(\mathcal{K}, \mathcal{X}, \mathcal{Y}, \mathcal{Z}, \mathcal{D})$ means that $\mathcal{K}$ is a list representing some natural number, $\mathcal{X}$, $\mathcal{Y}$ and $\mathcal{Z}$ are ground terms, $\mathcal{D}$ is a derivation tree for the ground atom $\mathrm{r}(\mathcal{X}, \mathcal{Y}, \mathcal{Z})$ and the height of this tree does not exceed the number represented by $\mathcal{K}$. A unary predicate symbol n will be additionally introduced with $\mathrm{n}(\mathcal{K})$ meaning that $\mathcal{K}$ is a list representing some natural number.

Taking into account what has been just said, one can easily see the semantical correctness of the program below, as well as the fact that a success of this program on the next goal can be immediately used to derive in **S** two ground atoms $\mathcal{A}$ and $\mathcal{B}$ such that ?– $\mathcal{A}, \mathcal{B}$ is an instance of g (when comparing both programs, one surely will notice the transpositions in the bodies of the new clauses corresponding to l and m; these transpositions aim at increasing of the efficiency).

```
?- n(K),d(K,x,x,W,D),d(K,y,y,W,E).

d(K,z,z,x,i).
d(K,z,z,y,j).
d([K],V,X,f(X,Y,U,V),k(D,E)) :-  d(K,X,Z,Y,D),d(K,U,Z,V,E).
d([K],Y,U,f(X,Y,U,V),l(D,E)) :-  d(K,U,Z,V,E),d(K,X,Z,Y,D).
d([K],X,U,Z,m(D,E)) :-  d(K,Y,U,Z,E),d(K,X,U,Y,D).
n([]).
n([K]) :- n(K).
```

During the executing of this program on the goal above, a systematic search will be performed. Namely, the program will first try to satisfy the goal with K=[], then with K=[[]] and so on until possibly succeeds.

We executed the above program on the above goal using a computer with a 133 Mhz CPU and running Dimiter Dobrev's freeware Strawberry Prolog (can be downloaded from http://www.dobrev.com). The run time was about 40 minutes and the goal succeeded with the following substitution (the specified list for K indicates that the terms for D and E have heights not exceeding 5, and in fact the heights of both these terms are equal to 5):

K=[[[[[[]]]]]],
W=f(x,f(z,y,y,f(z,x,z,y)),y,f(z,x,x,f(z,y,z,x))),

D=m(l(i,k(j,i)),k(m(l(i,j),k(j,k(i,j))),m(l(j,i),k(i,k(j,i))))),
E=m(l(j,k(i,j)),l(m(l(i,j),k(j,k(i,j))),m(l(j,i),k(i,k(j,i))))).

Remark 1. In the case of using a better computer and a commercial Prolog implementation the run time could be shorter, however a fair comparison with such other Prolog implementations must concern only ones that, like Strawberry Prolog, have the "occurs check" feature.

Let us introduce short denotations for the complex subterms of the indicated term for W (including this term itself). We set

#1=f(z,x,z,y), #2=f(z,y,z,x), #3=f(z,y,y,#1), #4=f(z,x,x,#2), #5=f(x,#3,y,#4).

(#5 is the term for W). The success of the program on the goal with the specified substitution allows one to derive in **S** the ground atoms r(x,x,#5) and r(y,y,#5) (one uses the terms for D and E, respectively, for that purpose). Here is a linearization of the derivation of the first of these atoms (the axioms and the rules of **S** are denoted in the same way as the corresponding clauses of the first program and, in order to make the things more transparent, we have added after any line of the derivation the corresponding derivation tree):

```
 1  r(z,z,x)    by i
      i
 2  r(z,z,y)    by j
      j
 3  r(y,z,#1)   by 1,2,k
      k(i,j)
 4  r(x,z,#1)   by 1,2,l
      l(i,j)
 5  r(x,z,#2)   by 2,1,k
      k(j,i)
 6  r(y,z,#2)   by 2,1,l
      l(j,i)
 7  r(#2,z,#4)  by 1,5,k
      k(i,k(j,i))
 8  r(x,x,#4)   by 1,5,l
      l(i,k(j,i))
 9  r(#1,z,#3)  by 2,3,k
      k(j,k(i,j))
10  r(x,z,#3)   by 4,9,m
      m(l(i,j),k(j,k(i,j)))
11  r(y,z,#4)   by 6,7,m
      m(l(j,i),k(i,k(j,i)))
12  r(#3,x,#5)  by 10,11,k
      k(m(l(i,j),k(j,k(i,j))),m(l(j,i),k(i,k(j,i))))
13  r(x,x,#5)   by 8,12,m
      m(l(i,k(j,i)),k(m(l(i,j),k(j,k(i,j))),m(l(j,i),k(i,k(j,i)))))
```

The solution of Problem 5 obtained in this way is not essentially different from its solution extracted from the solution of Problem 4 given by MKRP and both solutions mirror the essence of Prodanov's solution of Problem 3. The solution has an imperfection that is present (up to inessential details) also in the two other solutions just mentioned (in Prodanov's solution the imperfection remains actually deeply hidden behind the used mathematical denotations). The imperfection can be described as using a greater number of auxiliary objects than it is necessary. Namely, it is possible to reduce the length and the complexity of the derivations by using only one of the ground terms #1 and #2, since their properties used in the derivations are one and the same, namely the derivability in **S** of the instances of r(x,z,W) and r(y,z,W) obtained by replacement of W with the considered ground term. A solution without the mentioned imperfection can be found by looking for other substitutions that satisfy the same goal.

Remark 2. Appropriate utilities can be added to the used Prolog program to make it able to produce automatically linearizations of the found derivations.

We think Problem 5 could be useful for testing the capabilities of proof search programs.

REFERENCES

1. Ohlbach, H. J. and Siekmann, J. H. The Markgraf Karl Refutation Procedure. In: *Computational Logic, Essays in Honor of Alan Robinson*, MIT Press, 1991 (ed. J. L. Lassez and G. Plotkin), 41–112 (cf. also http://www.pms.ifi.lmu.de/mitarbeiter/ohlbach/homepage/publications/PL/MKRP/MKRP.pdf).
2. Prenowitz, W. A contemporary approach to classical geometry. *Amer. Math. Monthly*, **68** (1961), No. 1, part II, vi+67pp.
3. Prodanov, I. Zweifachassoziative Räume. *Annuaire de l'Univ. de Sofia, Fac. de Math.*, **57** (1964), 393–422 (Bulgarian. German summary).
4. Tagamlitzki, Y. Über die Trennbarkeitsbedingungen in den abelschen assoziativen Räumen. *Acad. des Sci. de Bulgarie, Bull. de l'Inst. de Math.*, **7** (1963), 169–183 (Bulgarian. Russian and German summaries).